
Mail Merge
Performing a Mail Merge in VBscript
The following example performs a mail merge entirely using VBscript.  This example is fairly complex and utilizes some previously undocumented functions.
Overview
This example uses the MergeFromFile function.  This function is not officially part of the SDK and is not supported as it may change in future releases.  It is believed that this function is correct as it is used by the SalesLogix Client to implement the mail merge functionality.  SalesLogix will make every attempt to resolve issues with this function.
This example also uses the SalesLogix document definition (.SDD) file format.  This document format is not available but can be manipulated through the SLXDoc object.  No additional documentation on the SDD format, the SLXDoc object, or how mail merge templates are stored is planned.  Creating and updating mail merge templates must be done through the SalesLogix interface.
The following example uses the SalesLogix Eval database to perform a mail merge with output to file for the contacts Jim Berghold, John Contorno, and Willis Wilsey.
Example Code

'Including Script - System:SLX Error Support

Option Explicit

Sub Main

    On Error Resume Next

    ' AttachmentType

    Const atRegularAttachment = 0 ' Default

    Const atLibraryAttachment = 1

    Const atTempAttachment = 2

    ' EmailFormat

    Const formatHTML = 0 ' Default

    Const formatPlainText = 1

    ' EmailSystem

    Const emailNone = 0

    Const emailExchange = 1

    Const emailOutlook = 2 ' Default

    Const emailSalesLogix = 3

    ' FaxDelivery

    Const fdASAP = 0 ' Default

    Const fdDateTime = 1

    Const fdOffPeak = 2

    Const fdHold = 3

    ' FaxPriority

    Const fpHigh = 0

    Const fpNormal = 1 ' Default

    Const fpLow = 2

    ' FollowUp

    Const fuNone = 0 ' Default

    Const fuMeeting = 1

    Const fuPhoneCall = 2

    Const fuToDo = 3

    ' GroupFamily

    Const famContact = 0 ' Default

    Const famAccount = 1

    Const famOpportunity = 2

    ' MergeWith

    Const withCurrentContact = 0

    Const withCurrentGroup = 1

    Const withSpecificGroup = 2

    Const withAccount = 3

    Const withOpportunity = 4

    Const withContactIDs = 5 ' Default

    ' OutputType

    Const otPrinter = 0

    Const otFile = 1

    Const otEmail = 2 ' Default

    Const otFax = 3

    ' TransportType

    Const transHTTP = 0

    Const transNative = 1

    Const transSockets = 2 ' Not Used

    Const transUndefined = 3 ' Default

    Dim oSLXDocument

    Dim oFSO

    Dim oFile

    Dim vFileName

    Dim vLibraryPath

    Dim vSecondsPerDay

    Dim vDateTime

    Dim vDuration

    ' This file will be deleted if the merge is successful. It's a temporary file only. If there

    ' were errors this file may remain (e.g. when the engine doesn't get called), so delete it

    ' if it is found.
    vFileName = "C:\Example.sdd"
    Set oFSO = CreateObject("Scripting.FileSystemObject")
    If ErrorCheck("There was an error creating the Scripting.FileSystemOject object.") Then Exit Sub
    If oFSO.FileExists(vFileName) Then

      Set oFile = oFSO.GetFile(vFileName)

      oFile.Delete()

    End If
    Set oFile = Nothing

    Set oFSO = Nothing
    If ErrorCheck("There was an error using the Scripting.FileSystemOject object.") Then Exit Sub
    Set oSLXDocument = CreateObject("SLXDoc.SLXDocument")
    If ErrorCheck("Error creating the SalesLogix Document Description object.") Then Exit Sub
    If oSLXDocument.CreateNew(vFileName) Then

       ' ---------------------------------------------------------------------------------------------
       ' Base options

       ' Set internally based on system settings

       ' oSLXDocument.MailMergeInformation.AttachmentPath = ""

       ' oSLXDocument.MailMergeInformation.BaseKeyCode = ""

       ' oSLXDocument.MailMergeInformation.BaseTable = ""

       ' oSLXDocument.MailMergeInformation.ConnectionString = ""

       ' oSLXDocument.MailMergeInformation.Remote = ""

       ' oSLXDocument.MailMergeInformation.SiteCode = ""

       ' oSLXDocument.MailMergeInformation.TransportType = transNative

       ' oSLXDocument.MailMergeInformation.UserID = ""
       ' True to attach the merged document to the contact

       oSLXDocument.MailMergeInformation.DoAttachments = True
       ' True to use contact's solicitation options (DONOTSOLICT, DONOTEMAIL, DONOTFAX, etc.)

       oSLXDocument.MailMergeInformation.DoNotSolicit = True
       ' If MergeSilently is True and EditAfter is True there will be an error

       ' since EditAfter requires interation with the UI.

       oSLXDocument.MailMergeInformation.EditAfter = False
       ' If MergeSilently is True and EditBefore is True there will be an error

       ' since EditBefore requires interation with the UI.

       oSLXDocument.MailMergeInformation.EditBefore = False
       ' MergeSilently is currently forced to False internally. It will be fixed in SalesLogix 6.1.

       oSLXDocument.MailMergeInformation.MergeSilently = False
       ' ---------------------------------------------------------------------------------------------
       ' The type of merge output: otPrinter, otFile, otEmail, or otFax.

       'oSLXDocument.MailMergeInformation.OutputTo = otEmail

       'oSLXDocument.MailMergeInformation.OutputTo = otPrinter

       oSLXDocument.MailMergeInformation.OutputTo = otFile

       'oSLXDocument.MailMergeInformation.OutputTo = otFax
       ' ---------------------------------------------------------------------------------------------
       ' A type 25 plugin ID (Microsoft Word template)
       ' Need to change pluginID and name for real code
       oSLXDocument.MailMergeInformation.TemplatePluginID = "pQF8AA0001L6"

       ' The name of the TemplatePluginID (not required)

       oSLXDocument.MailMergeInformation.TemplatePluginName = "Prospect Email - Want to Get Away?"
       ' The user's name (not required)

       oSLXDocument.MailMergeInformation.UserName = Application.BasicFunctions.SystemInfoFor("UserName")
       ' ---------------------------------------------------------------------------------------------
       Select Case oSLXDocument.MailMergeInformation.OutputTo

         Case otEmail

           ' E-mail options

           ' Set internally based on system settings:

           ' oSLXDocument.MailMergeInformation.EmailClient = emailOutlook

           ' oSLXDocument.MailMergeInformation.LibraryPath = ""

           ' oSLXDocument.MailMergeInformation.MAPIProfileName = ""
           ' Semicolon delimited list of e-mail addresses and/or display names

       
' (e.g. 'John Doe;mcessna@saleslogix.com').

           oSLXDocument.MailMergeInformation.EmailBCC = ""
           ' Semicolon delimited list of e-mail addresses and/or display names

' (e.g. 'John Doe;mcessna@saleslogix.com').

           oSLXDocument.MailMergeInformation.EmailCC = ""
           ' formatHTML or formatPlainText

           oSLXDocument.MailMergeInformation.EmailFormat = formatHTML
           ' EmailFrom is only used if the e-mail is to be sent on behalf of you by using someone elses address

           oSLXDocument.MailMergeInformation.EmailFrom = ""
           ' True to save a copy of the message in the user's Sent box.

           oSLXDocument.MailMergeInformation.EmailSaveCopy = False
           ' The subject of the e-mail.

           oSLXDocument.MailMergeInformation.EmailSubject = "E-mail Subject"
           ' OverrideAttachments must be set to True if you add attachments using oSLXDocument.AddAttachment()
           ' Using AddAttachment() will override the attachments associated with a template.

           oSLXDocument.MailMergeInformation.OverrideAttachments = False
         Case otFax

           ' FAX options are used for an OutputTo of otFax

           oSLXDocument.MailMergeInformation.FaxBillingCode = ""
           ' The FaxClientCode property is not used because of WinFax API.

           oSLXDocument.MailMergeInformation.FaxClientCode = ""
           ' Value must be localized for 'Quick Cover Page' only.

           ' FaxCoverPage should be the cover page name from the WinFax cover page database.

           oSLXDocument.MailMergeInformation.FaxCoverPage = "Quick Cover Page"
           ' Used for a FaxDelivery of fdDateTime.

           oSLXDocument.MailMergeInformation.FaxDate = Now()

           oSLXDocument.MailMergeInformation.FaxDelivery = fdASAP

           oSLXDocument.MailMergeInformation.FaxKeywords = ""

           oSLXDocument.MailMergeInformation.FaxMessage = "Fax Message"

           oSLXDocument.MailMergeInformation.FaxPriority = fpNormal
           ' The name of the device you want to use for this send job.

           ' Note: 'Concord Internet Fax' will NOT work via the WinFax API.

           ' Leave FaxSendBy undefined to use the default Fax device.

           oSLXDocument.MailMergeInformation.FaxSendBy = ""
           ' The FaxSendSecure property is not used because of WinFax API.

           oSLXDocument.MailMergeInformation.FaxSendSecure = False

           oSLXDocument.MailMergeInformation.FaxSubject = "Fax Subject"
         Case otFile

           ' FileDirectory is used for an OutputTo of otFile

           ' GetPersonalDataPath = 'C:\Documents and Settings\%User%\My Documents\SalesLogix'.

           oSLXDocument.MailMergeInformation.FileDirectory = Application.BasicFunctions.GetPersonalDataPath
         Case otPrinter

           ' PrinterName is used for an OutputTo of otPrinter

           ' PrinterName is required

           oSLXDocument.MailMergeInformation.PrinterName = \\logixprint\LaserDEV
           ' Label options

           ' LabelID and LabelPrinter are used for an OutputTo of otPrinter when DoPrintLabels is True

           ' If MergeSilently is True and DoPrintLabels is True there will be an error

           ' since DoPrintLables requires interation with the UI.

           oSLXDocument.MailMergeInformation.DoPrintLabels = False
           ' Avery 5161

           oSLXDocument.MailMergeInformation.LabelID = "pQF8AA0003OY"
           ' If LabelPrinter is not defined the default printer will be used.

           oSLXDocument.MailMergeInformation.LabelPrinter = "\\logixprint\LaserDEV"

       End Select
       ' ---------------------------------------------------------------------------------------------
       ' MergeWith defines what entity type property the engine should use when merging

       ' (i.e. AccountID, OpportuntiyID, CurrentContactID, GroupID, or ContactIDs)
       'oSLXDocument.MailMergeInformation.MergeWith = withCurrentContact

       'oSLXDocument.MailMergeInformation.MergeWith = withSpecificGroup

       'oSLXDocument.MailMergeInformation.MergeWith = withAccount

       'oSLXDocument.MailMergeInformation.MergeWith = withOpportunity

       oSLXDocument.MailMergeInformation.MergeWith = withContactIDs
       Select Case oSLXDocument.MailMergeInformation.MergeWith

         Case withCurrentContact

           oSLXDocument.MailMergeInformation.CurrentContactID = "CGHEA0002670"

           oSLXDocument.MailMergeInformation.CurrentContactName = "John Abbott" ' not required
         Case withCurrentGroup, withSpecificGroup

           oSLXDocument.MailMergeInformation.GroupID = "GROUP0ALLACC"

           oSLXDocument.MailMergeInformation.GroupName = "All Accounts" ' not required

           oSLXDocument.MailMergeInformation.GroupFamily = famAccount ' not required'
         Case withAccount

           oSLXDocument.MailMergeInformation.AccountID = "AGHEA0002669"

           oSLXDocument.MailMergeInformation.AccountName = "Abbott Ltd." ' not required
         Case withOpportunity

           oSLXDocument.MailMergeInformation.OpportunityID = "OGHEA0002815"

           oSLXDocument.MailMergeInformation.OpportunityName = "Abbott Ltd.-Phase I"
         Case withContactIDs

           ' ContactIDs is a comma-delimited list of contact IDs.

           ' "Jim Berghold", "John Contorno", and "Willis Wilsey".

           oSLXDocument.MailMergeInformation.ContactIDs = "CA2EK0013116,CA2EK0013117,CA2EK0013118"

       End Select
       ' ---------------------------------------------------------------------------------------------
       ' History options

       oSLXDocument.MailMergeInformation.DoHistory = True
       If oSLXDocument.MailMergeInformation.DoHistory Then

          oSLXDocument.MailMergeInformation.HistoryInfoCategory = "History Category"

          oSLXDocument.MailMergeInformation.HistoryInfoNotes = "History Notes"

          oSLXDocument.MailMergeInformation.HistoryInfoRegarding = "History Regarding"

          oSLXDocument.MailMergeInformation.HistoryInfoResult = "History Result"

       End If
       ' ---------------------------------------------------------------------------------------------
       ' Schedule Follow-Up options are used when DoScheduleFollowUp is True

       oSLXDocument.MailMergeInformation.DoScheduleFollowUp = False
       If oSLXDocument.MailMergeInformation.DoScheduleFollowUp Then

          vDateTime = Now()

          vDuration = 15 ' minutes

          vSecondsPerDay = 24 * 60 * 60

          ' Minutes: vDateTime - ((vDuration * 60) / vSecondsPerDay) Hours: vDateTime - ((vDuration * 3600) / vSecondsPerDay)

          oSLXDocument.MailMergeInformation.ScheduleFollowUpAlarmTime = vDateTime - ((vDuration * 60) / vSecondsPerDay)

          oSLXDocument.MailMergeInformation.ScheduleFollowUpCarryOverNotes = True

          oSLXDocument.MailMergeInformation.ScheduleFollowUpCategory = "Follow-Up Category"

          oSLXDocument.MailMergeInformation.ScheduleFollowUpDuration = vDuration

          oSLXDocument.MailMergeInformation.ScheduleFollowUpNotes = "Follow-Up Notes"

          oSLXDocument.MailMergeInformation.ScheduleFollowUpRegarding = "Follow-Up Regarding"

          oSLXDocument.MailMergeInformation.ScheduleFollowUpSetAlarm = False

          oSLXDocument.MailMergeInformation.ScheduleFollowUpStartDate = vDateTime

          oSLXDocument.MailMergeInformation.ScheduleFollowUpTimeless = True

          oSLXDocument.MailMergeInformation.ScheduleFollowUpType = fuMeeting

          oSLXDocument.MailMergeInformation.ScheduleFollowUpUserID = Application.BasicFunctions.CurrentUserID

       End If
       ' ---------------------------------------------------------------------------------------------
       ' Summary Info (not required)

       oSLXDocument.SummaryInformation.Author = "Michael Cessna"

       oSLXDocument.SummaryInformation.Title = "Test of MergeFromFile()"
       ' ---------------------------------------------------------------------------------------------
       ' Commit the changes to the oSLXDocument

       oSLXDocument.Commit()
       ' ---------------------------------------------------------------------------------------------
       ' Close the oSLXDocument

       oSLXDocument.Close()
       ' ---------------------------------------------------------------------------------------------
       Set oSLXDocument = Nothing
       ' ---------------------------------------------------------------------------------------------
       ' Run the merge job

       If Not Application.BasicFunctions.MergeFromFile(vFileName) Then

         MsgBox "MergeFromFile failed or was canceled."

       End If
    Else

      Set oSLXDocument = Nothing

      MsgBox "There was an error."

    End If
    If ErrorCheck("There was an error.") Then Exit Sub
End Sub
